

Excel Funktionen

Kategorie Nachschlagen & Verweisen



INDEX-Funktion

- ▶ Als Matrix- oder Bezugsversion
 - Wir behandeln vorerst nur die Matrixversion, die genau **einen** Zellinhalt zurückgibt
- ▶ Argumente
 - Zellbereich
 - Zeilenindex (bezogen auf den Zellbereich)
 - Spaltenindex (bezogen auf den Zellbereich)



Anweisung	Ergebnis
=Index(B2:E8; 2; 3)	Liefert den Wert von D3 (innerhalb des Bereiches liegt die Zelle D3 am Kreuzungspunkt der 2. Zeile und der 3. Spalte)

VERGLEICH-Funktion

- ▶ Durchsucht einen Bereich nach einem angegebenen Suchbegriff
- ▶ Funktionswert ist die relative Position der Zelle innerhalb des Bereiches, die den gesuchten Wert enthält
- ▶ Drei Argumente
 - Suchbegriff
 - Zellbereich (einspaltig oder einzeilig)
 - Vergleichstyp
- ▶ Wird der gesuchte Eintrag nicht gefunden, so ist der Funktionswert der Fehler #NV


VERGLEICH-Funktion

- ▶ Vergleichstyp: 0 (Genaue Übereinstimmung)
 - Zellbereich muss nicht sortiert sein

Zellbereich	Anweisung	Ergebnis
	=Vergleich("Sakko"; A1:A4; 0)	3 (Suchbegriff Sakko in der 3. Zelle des Zellbereichs)
	=Vergleich(20; A1:A4; 0)	2 (Suchbegriff 20 in der 2. Zelle der Zellbereichs)
	=Vergleich(15; A1:A4; 0)	#NV (Suchbegriff 15 nicht gefunden)


VERGLEICH-Funktion

- ▶ Vergleichstyp: 1 (Kleiner als)
 - Grösster Wert, der kleiner oder gleich dem Suchbegriff ist
 - Meistens bei numerischen Werten
 - Zellbereich muss **aufsteigend** sortiert sein

Zellbereich	Anweisung	Ergebnis
	=Vergleich(9; A1:A4; 1)	2 (5 ist der grösste Wert kleiner oder gleich dem Suchbegriff 9)
	=Vergleich(2; A1:A4; 1)	#NV (Suchbegriff 2 ist kleiner als der erste Wert im Zellbereich)
	=Vergleich(5; A1:A4; 1)	2 (5 ist der grösste Wert kleiner oder gleich dem Suchbegriff 5)
	=Vergleich(30; A1:A4; 1)	4 (27 ist der grösste Wert kleiner oder gleich dem Suchbegriff 30)

VERGLEICH-Funktion

- ▶ Vergleichstyp: -1 (Grösser als)
 - Kleinster Wert, der grösser oder gleich dem Suchbegriff ist
 - Meistens bei numerischen Werten
 - Zellbereich muss **absteigend** sortiert sein

Zellbereich	Anweisung	Ergebnis
	=Vergleich(9; A1:A4; -1)	2 (20 ist der kleinste Wert grösser oder gleich dem Suchbegriff 9)
	=Vergleich(30; A1:A4; -1)	#NV (Suchbegriff 30 ist grösser als der erste Wert im Zellbereich)
	=Vergleich(5; A1:A4; -1)	3 (5 ist der kleinste Wert grösser oder gleich dem Suchbegriff 5)
	=Vergleich(2; A1:A4; -1)	4 (3 ist der kleinste Wert grösser oder gleich dem Suchbegriff 3)

Kombination INDEX und VERGLEICH

- ▶ Ermitteln der verkauften Menge an Westen in Schwaz

	A	B	C	D	E
1	Filiale/Artikel	Hemd	Sakko	Weste	Hose
2	Innsbruck	32	32	23	26
3	Schwaz	17	40	19	13
4	Hall	11	39	18	34

- ▶ Ermitteln der Positionen von "Weste" und "Schwaz"
 =Vergleich ("Schwaz"; A2:A4; 0) → 2
 =Vergleich("Weste"; B1:E1; 0) → 3
- ▶ Kombination
 =Index(B2:E4; Vergleich("Schwaz"; A2:A4; 0);
 Vergleich("Weste"; B1:E1; 0))

Kombination INDEX und VERGLEICH

- ▶ Rabatt beim Kauf von 22 Westen

	A	B	C	D	E
1	Menge/Artikel	Hemd	Sakko	Weste	Hose
2	0	0%	0%	0%	0%
3	20	10%	8%	4%	7%
4	40	20%	16%	12%	14%

- ▶ Ermitteln der Positionen von 22 und "Weste"
 =Vergleich (22; A2:A4; 1) → 2
 =Vergleich("Weste"; B1:E1; 0) → 3
- ▶ Kombination
 =Index(B2:E4; Vergleich(22; A2:A4; 1);
 Vergleich("Weste"; B1:E1; 0))

SVERWEIS-Funktion

- ▶ Durchsucht die **erste** Spalte eines Bereichs und liefert bei Übereinstimmung den Inhalt der Zelle einer angegebenen Spalte
- ▶ Beispiel
 - Gesucht wird der Preis des Artikels mit der Artikelnummer A38IU

	A	B	C
1	ArtNr	Artikel	Preis
2	A01XY	Festplatte	€ 100,00
3	B45PA	Maus	€ 25,00
4	A38IU	Tastatur	€ 38,00
5	Z78EW	Bildschirm	€ 350,00

SVERWEIS-Funktion

- ▶ Vier Argumente
 - Suchbegriff
 - Zellbereich
 - Hinweis: Werte, die mit dem Suchbegriff abgeglichen werden, müssen sich in der **ersten** Spalte befinden
 - Spaltenindex
 - Zahl zwischen 1 und Anzahl der Spalten des Zellbereichs
 - Bereich_Verweis
 - FALSCH
 - Genaue Übereinstimmung
 - Werte in erster Spalte des Zellbereichs müssen nicht sortiert sein
 - WAHR
 - Als Treffer gilt der grösste Wert, der kleiner oder gleich dem Suchbegriff ist (siehe Vergleichstyp 1 bei Funktion Vergleich)
 - Werte in erster Spalte des Zellbereichs müssen aufsteigend sortiert sein

SVRWEIS-Funktion

	A	B	C
1	ArtNr	Artikel	Preis
2	A01XY	Festplatte	€ 100,00
3	B45PA	Maus	€ 25,00
4	A38IU	Tastatur	€ 38,00
5	Z78EW	Bildschirm	€ 350,00

Anweisung (Genaue Übereinstimmung)	Ergebnis
=SVerweis("A38IU"; A2:C5; 1; FALSCH)	A38IU
=SVerweis("A38IU"; A2:C5; 2; FALSCH)	Tastatur
=SVerweis("A38IU"; A2:C5; 3; FALSCH)	38
=SVerweis("A38IU"; A2:C5; 4; FALSCH)	#NV (Zellbereich umfasst nur drei Spalten)
=SVerweis("A38IUZ"; A2:C5; 2; FALSCH)	#NV (Suchbegriff nicht gefunden)

SVRWEIS-Funktion

	A	B	C
1	Gehalt	Gruppe	Voraussetzung
2	0,00	A01	Keine
3	1.800,00	A02	Pflichtschule
4	2.500,00	A03	Matura
5	2.800,00	A04	Studium

Anweisung (Ungefähre Übereinstimmung)	Ergebnis
=SVerweis(1900; A2:C5; 1; WAHR)	1.800,00
=SVerweis(1900; A2:C5; 2; WAHR)	A02
=SVerweis(3000; A2:C5; 2; WAHR)	A04
=SVerweis(-100; A2:C5; 2; WAHR)	#NV (Suchbegriff kleiner als der kleinste Wert)
=SVerweis(1900; A2:C5; 2; FALSCH)	#NV (Genauere Übereinstimmung; 1900 ist nicht vorhanden)

SVRWEIS-Funktion: #NV

- ▶ Findet die SVRWEIS-Funktion keine Übereinstimmung, so wird der Fehlerwert #NV geliefert
- ▶ In den Unterlagen zur Kategorie Logik ist beschrieben, wie man diese Fehlerwerte abfängt
 - Kombination von WENN- und ISTFHLER-Funktion
=Wenn (IstFehler(SVerweis(.....)); "Nicht gefunden";
SVerweis (.....))
 - WENNFEHLER-Funktion
=WennFehler(SVerweis (.....); "Nicht gefunden")

Kombination SVERWEIS und VERGLEICH

- ▶ Ermitteln der verkauften Menge an Westen in Schwaz

	A	B	C	D	E
1	Filiale/Artikel	Hemd	Sakko	Weste	Hose
2	Innsbruck	32	32	23	26
3	Schwaz	17	40	19	13
4	Hall	11	39	18	34

- ▶ Ermitteln der Positionen von "Weste"

=Vergleich("Weste"; B1:E1; 0) → 3

- ▶ Kombination *)

=SVerweis("Schwaz"; A1:E4; Vergleich("Weste"; B1:E1; 0) +1;
FALSCH)

*) Ergebnisse von Vergleich(...) um 1 erhöhen, damit die richtige Position als Spalte in der SVerweis-Funktion verwendet wird.

Kombination SVERWEIS und VERGLEICH

- ▶ Rabatt beim Kauf von 22 Westen

	A	B	C	D	E
1	Menge/Artikel	Hemd	Sakko	Weste	Hose
2	0	0%	0%	0%	0%
3	20	10%	8%	4%	7%
4	40	20%	16%	12%	14%

- ▶ Ermitteln der Positionen von "Weste"
=Vergleich("Weste"; B1:E1; 0) → 3
- ▶ Kombination *)
=SVerweis(22; A1:E4; Vergleich("Weste"; B1:E1; 0) + 1;
WAHR)

*) Ergebnisse von Vergleich(...) um 1 erhöhen, damit die richtige Position als Spalte in der SVerweis-Funktion verwendet wird.

WVERWEIS-Funktion

- ▶ Prinzipiell gleich wie die SVERWEIS-Funktion ^{*)}, durchsucht aber die **erste Zeile** eines Bereichs und liefert bei Übereinstimmung den Inhalt der Zelle einer angegebenen **Zeile**
- ▶ Beispiel
 - Gesucht wird der Inhalt der dritten Zeile im Bereich A1:C5 aus der Spalte Artikel: =WVerweis("Artikel"; A1:C5; 3; FALSCH)

	A	B	C
1	ArtNr	Artikel	Preis
2	A01XY	Festplatte	€ 100,00
3	B45PA	Maus	€ 25,00
4	A38IU	Tastatur	€ 38,00
5	Z78EW	Bildschirm	€ 350,00
-			

^{*)} Probieren Sie selbst die Funktion mit unterschiedlichen Werten und Einstellungen.